Problem:
There are n processes in a queue. Each process has namei and timei. The round-robin scheduling handles the processes in order. A round-robin scheduler gives each process a quantum (a time slot) and interrupts the process if it is not completed by then. The process is resumed and moved to the end of the queue, then the scheduler handles the next process in the queue.

For example, we have the following queue with the quantum of 100ms.

A(150) - B(80) - C(200) - D(200)

First, process A is handled for 100ms, then the process is moved to the end of the queue with the remaining time (50ms).

B(80) - C(200) - D(200) - A(50)

Next, process B is handled for 80ms. The process is completed with the time stamp of 180ms and removed from the queue.

C(200) - D(200) - A(50)

Your task is to write a program which simulates the round-robin scheduling.

Input

n q

name1 time1

name2 time2

...

namen timen

In the first line the number of processes n and the quantum q are given separated by a single space.

In the following n lines, names and times for the n processes are given. namei and timei are separated by a single space.

Output

For each process, prints its name and the time the process finished in order.

Constraints

1 ≤ n ≤ 100000

1 ≤ q ≤ 1000

1 ≤ timei ≤ 50000

1 ≤ length of namei ≤ 10

1 ≤ Sum of timei ≤ 1000000

Sample Input 1

5 100
p1 150
p2 80
p3 200
p4 350
p5 20

Sample Output 1

p2 180
p5 400
p1 450
p3 550
p4 800

Notes

Template in C

Solution in Haskell (judge verdict: Accepted)

import Control.Applicative ((<$>))
import Control.Monad (unless)
import Data.ByteString.Char8 (ByteString)
import qualified Data.ByteString.Char8 as B

import Control.Monad.State

main :: IO ()
main = do
  [_, q] <- map read <$> words <$> getLine
  que <- Queue [] <$> map (f . B.words) <$> B.lines <$> B.getContents
  solve q que
  where f [n, t] = (n, readi B.readInt t)
  
solve :: Int -> Queue (ByteString, Int) -> IO ()
solve q que = evalStateT (rrs 0) que
  where rrs :: Int -> StateT (Queue (ByteString, Int)) IO ()
        rrs ct = do
          que <- get
          unless (isEmptyQue que) $ do
            let (Just ((n, t), nque)) = deQue que
            if t > q then do
              put $ enQue (n, t - q) nque
              rrs (ct+q)
            else do
              lift $ putStrLn $ B.unpack n ++ " " ++ show (ct+t)
              put nque
              rrs (ct+t)

data Queue a = Queue [a] [a] deriving Show

enQue :: a -> Queue a -> Queue a
enQue x (Queue xs ys) = Queue (x:xs) ys

deQue :: Queue a -> Maybe (a, Queue a)
deQue (Queue [] []) = Nothing
deQue (Queue xs (y:ys)) = Just (y, Queue xs ys)
deQue (Queue xs []) = deQue (Queue [] (reverse xs))

isEmptyQue :: Queue a -> Bool
isEmptyQue (Queue [] []) = True
isEmptyQue _ = False

readi :: Integral a =>  (ByteString -> Maybe (a, ByteString)) -> ByteString -> a
readi f s = let Just (n, _) = f s in n